
 

 

 

 

 

 

Welcome to “Budgeting & Forecasting in Excel”! This 
handbook contains some take-away instructions for some of the 
practical exercises we’ll cover during the course. You do not need 
to read it in advance, but you might like to print it for easy 
reference. The course materials also contain some references to 
the two books by Danielle Stein Fairhurst “Using Excel for 
Business and Financial Modelling” and “Financial Modeling in 
Excel for Dummies”.  These books can be purchased online from 
Amazon or may be included in your enrolment. It is a good 
accompanying reference for the course, but it not required that 
you have a copy when taking the course. 

FORECASTING FOR SEASONALITY 

 
 
Seasonality can be stripped out of historical numbers, and then reapplied to 
provide a more accurate forecast. 
 
See page 158 of Using Excel for Business and Financial Modelling, for more 
information on how to use the FORECAST function. 

 

Based on the following data, create a budget for the four quarters of 2021. 
Remove the quarter-related seasonality, forecast for 2021 using the FORECAST 
function, and then reapply seasonality. (Assume a calendar year) 
 

 Q1 Q2 Q3 Q4 
2018 520 730 820 530 
2019 590 810 900 600 
2020 650 900 1000 650 

 
1. Enter this table into a blank Excel workbook, starting from cell A1, or use the template 

provided. 
2. Firstly, we need to calculate the seasonality for each quarter.   Below the table in the next 

row, calculate the average for each quarter.   
3. Insert the formula =AVERAGE(B3:B5) in cell B6 and copy it across.  You will notice a trend 

– some quarters are higher and lower than others. 
1. Now to calculate the seasonality index we need to compare the average of the quarter to 

the average of the data as a whole.  Your formula in cell B7 should be 
=B6/AVERAGE($B$3:$E$5), and the answer is 0.81. 

2. Copy this across, and sum all of the seasonality indexes together to check.  The total should 
be 4. 

4. Your model should look something like this: 

 

5. The FORECAST function needs the numbers to be in a consecutive line, so create a table 
further down from cell A11 with the exact quarter end date beginning with 31st March 
2018.   

6. Begin the next quarter by adding 3 months to the previous cell with an EDATE function. 
Your formula in cell A12 should be =EDATE(A11,3) Copy it down, through to the end of 
December 2021.  

Note that because 2020 is a leap year, the dates in 2020 are a day earlier 
than in the previous years, but the EDATE (or the EOMONTH function) will take 
care of that. 
See page 165 of Using Excel for Business and Financial Modelling, for more 
information on Working with Dates 

1. In column B, transpose the data from the table above using either a paste values transpose 
as shown below, or a =TRANSPOSE array formula. 

 

2. Cut and paste the data until it looks like the below: 

 
 
Now, try using the FORECAST function in cells B23 to B26 and then graph it.  Use 
the FORECAST function.   

 X is the date or number for which you want the numbers to be forecast.   
 Known Y’s is the known historical data 
 Known X’s is the known historical dates or numbers 

Note, the TREND formula will also work in pretty much the same way.   
Your formula in cell B23 should be:  
=FORECAST(A23,$B$11:$B$22,$A$11:$A$22) 
Create a line graph of your data which should look like this: 

 
Note that you may need to change the formatting of the dates on the X-Axis by 
double-clicking on the X-Axis and changing the format in the panel which 
appears to the right as shown below: 

What we have forecast so far will evenly smooth the quarters and apply the 
linear trend, but it will not show the seasonality.  In order to forecast these 
numbers more accurately, we need to apply the seasonality. 

3. Transpose the seasonality numbers from cells B7:E7 to cells C23:C26.  You may link, or 
hard-code, depending on whether you want the model to be dynamic or static.  If you 
have the time, a dynamic solution is always best! 

4. Leave the rest of the fields blank. 
5. Multiply the seasonality index by the raw data to achieve the seasonalised data.  Your 

sheet should look something like this: 

6. Add another series to the chart such that it looks like this: 

Tip: you can do this by editing the chart, or simply copy the data in cells D11 to 
D26 and paste it directly into the chart. 
Forecast Sheet: Note that instead of using this method, simply highlight the 
historical data, and go to Data – Forecast Sheet and the forecasted seasonality 
with three confidence intervals are automatically created for you as shown 
below: 

MODELLING FOR STEPPED COSTS 

 

ROUND Functions 

There are several functions in Excel which round off numbers.  The ones that 
are the most useful for modellers are ROUND, ROUNDUP and ROWNDOWN.  
Each of these work in exactly the same way, except that the ROUND function 
will round to the nearest specified value; whether it is up or down.  The 
ROUNDUP function will round up to the nearest, and – you guessed it – the 
ROUNDDOWN will round down to the nearest. 
To understand how this particular function works, instead of showing you a 
pre-worked example, it’s best to simply try it out for yourself by following 
these steps: 

1. Open up Excel, and in cell A1 enter a long number with several decimal places, such 
as 45215754.575 

2. On the Home tab, in the Numbers section, press the comma button once to format the 
cell.  The formatting will change so the number now looks like this: 45,215,754.58.  
Note that the number remains the same; the third decimal place is still there but it’s 
just not showing due to the way the cell is now formatted. 

3. In cell A2, enter the formula =ROUND(A1,1).  This will reduce the decimal places to 
only one so that cell A2 contains the value 45,215,754.60.  Note that the number 
really does only contain one decimal place; the extra decimal places are not being 
suppressed by formatting as they are in cell A1. 

4. In cell A3, try removing all decimal places completely with the formula 
=ROUND(A1,0) 

5. Now in cell A4, we’ll go one step further and round our number to the nearest ten 
with the formula =ROUND(A1,-1).  The result will be 45,215,750.00 

6. Then in cell A5, try rounding it to the nearest thousand with =ROUND(A1,-3).  The 
result will be 45,216,000.00 

7. And finally in cell A6, round to the nearest million with =ROUND(A1,-6). The result 
will be 45,000,000.00, as shown in figure 7-17. 

 
 
Note that the ROUNDUP and ROUNDDOWN function will work in exactly the 
same way, except to round up or down instead of the nearest. 

Figure 7-17 Using the ROUND function 

Forecasting Fixed, Variable and Stepped Costs with ROUNDUP 

For a practical example of how to use the ROUNDUP function as part of a 
financial model, follow this series of steps as shown.  

 Let’s say you are building a five-year plan for a call centre 
 Each call operator can handle 40 customers. 
 In year 1 you have 500 customers, and you expect that to increase by 

20% per year 
 How many call operators will you need each year?  Hint: don’t forget to 

ROUNDUP! 
 You have fixed costs of $250,000 
 Variable overheads are $100 per customer  
 Each call operator costs $65,000 per annum 
 All costs increase by inflation of 3% 

This exercise demonstrates the difference between fixed, variable and 
stepped costs, which are important concepts in management accounting and 
budget modelling.  As shown in figure 7-18, fixed costs do not change as the 
number of units produced increases.  Variable costs change directly in line with 
the number of units produced, and are also very simple to model.  Stepped 
costs, however, do not increase in a linear fashion, but increase in increments 
or “steps” and these can be modelled using the ROUNDUP function. 
 
 
 

Figure 7-18 Fixed, Variable and Stepped Costs 
This exercise also demonstrates a common escalation technique for increasing 
amounts by a growth rate, or inflation amount. We often need to be able to 
include escalation in our models for the purpose of forecasting such as our 
example here. In this exercise, we need to increase our number of customers 
by 20% each year.  The number of customers in the first year is 500.  When 
we add the 20% growth to this number it becomes 600.  We then calculate 
the customer number in the third year based on this number (600) not the first 

year (500). This effect whereby the growth increases exponentially is called 
compounding. 
To calculate growth, we use the formula: 
base amount*(1+growth) 
The reason we are adding 1 (effectively 100 percent) to the growth rate 
(20%) is because we want the capital sum returned along with the accrued 
interest. Multiplying an amount by (1+growth) is a very common calculation in 
financial modelling. 

1. Download the sample file or enter and format data as shown in Figure 7-19 
2. Firstly, we need to calculate the number of customers over time.  Select cell C12 

and enter the formula =B12*(1+$B$4) to calculate the number of customers 
expected in Year 2. Don’t forget to use the F4 shortcut to anchor the cell 
reference. 

3. Copy this formula all the way across the row. The calculated result in Year 5 is 
1,037. 

4. Next, in cell B13 we need to work out how many call operators we need each 
year, so divide the number of customers by the customers per operator with the 
formula =B12/$B$3 and the calculated result is 12.50. Don’t forget to use the 
F4 shortcut to anchor the cell reference. 

5. Huh?  That doesn’t make sense because we can’t employ a fraction of a person! 
We’ll need to round up to the nearest whole person with the formula by 
wrapping a ROUNDUP function around our existing formula like this: 
=ROUNDUP(B12/$B$3,0) 

6. Copy it all the way across the row and the calculated result in Year 5 is 26.   
Compare your results to Figure 7-19. 

 
Figure 7-19 Calculating number of call operators required with ROUNDUP 
We now have calculated the number of staff required so we can use this later 
on in row 20 to work out their costs.  For now, we will continue working down 
the page. 

In row 14 we need to calculate the overhead amounts for each year. There 
are two parts to this calculation; firstly, we need to apply inflation, and 
secondly multiply it by the number of customers.  We could do the entire 
calculation in a single row, but the formula would be difficult to follow. 

<Remember> 
When it comes to formula layout in a financial model, simple is best!  It is far 
better to lay out the calculation step by step instead of trying to do the whole 
thing in one row. 

7. Select cell B14 and link this to the overhead assumption with the formula =B7 
8. In cell C14 we need to add inflation.  We’ll use exactly the same technique as 

we did when increasing the number of customers, but we’ll pick up the inflation 
assumption instead of the growth assumption.  Enter the formula =B14*(1+$B$5).   
Don’t forget to use the F4 shortcut to anchor the cell reference. 

9. Copy it all the way across the row and the calculated result in Year 5 is 113. 

Note that we have different formulas in cell B14 and C14.  Ordinarily, we try 
to have consistent formulas wherever possible, to reduce the number of 
formulas in the model but in this case, it’s not practical. 
In row 15 we need to calculate the staff costs per operator for each year.  
This works in exactly the same way as the overhead costs.  We’ll increase the 
per operator costs by inflation each year and multiply it by the number of 
operators later on. 

10. Select B15 and link this to the cost per operator with the formula =B8 
11. Add inflation to this number in cell C15 with the formula =B15*(1+$B$5) (don’t 

forget to F4!) 
12. Copy it all the way across the row and the calculated result in Year 5 is 

$73,158. 

Now we’ve finished the workings block, we can start to calculate the costs 
below in row 18. 

13. Select B18 and link this to the fixed cost assumption with the formula =B6 
14. Add inflation to this number in cell C18 with the formula =B18*(1+$B$5)  
15. Copy it all the way across the row and the calculated result in Year 5 is 

$281,377 
16. In B19, calculate the total overhead costs by multiplying the overhead cost per 

customer per year by the number of customers in that year with the formula 
=B14*B12. No need to lock the cell referencing, as we want this to copy across 
the row. 

17. Copy it all the way across the row and the calculated result in Year 5 is 
$116,693 

18. Next, in cell B20, calculate the total call staff costs by multiplying the cost per 
operator by the number of operators with the formula =B15*B13 

19. Copy it all the way across the row and the calculated result in Year 5 is 
$1,902,110 

20. Lastly, sum the total costs in cell B21 with the formula =SUM(B18:B20) or select 
cell B21 and use the shortcut ALT+= to add up all the costs. 

21. Copy it all the way across the row and the calculated result in Year 5 is 
$2,300,180.  Compare your results to figure 7-20. 

Figure 7-20 Completed Five Year Strategic Forecast 

BUILDING DROP-DOWN SCENARIOS 

The most commonly used method of building scenarios (and the one that I most 
often teach in my training courses) is to use a combination of formulas and 
drop-down boxes. In the model, you create a table of possible scenarios and 
their inputs and link the scenario names to an input cell drop-down box. The 
inputs of the model are linked to the scenario table. If the model has been 
built properly with all the inputs flowing through to the outputs, then the results 
of the model will change as the user selects different options from the drop-
down box. 
Data validation drop-down boxes are used for a number of different 
purposes in financial modeling, including scenario analysis.  

Using data validations to model profitability scenarios 

Go to the tab called “Stepped”. 
The way I’ve modeled this, the inputs are lined up in column B. You could 
perform sensitivity analysis simply by changing one of the inputs — for 
example, change the customers per call operator in cell B3 from 40 to 45, and 
you’ll see all the dependent number change. This would be a sensitivity 
analysis, because you’re changing only one variable. Instead, you’re going to 
change multiple variables at once in this full scenario analysis exercise, so 
you’ll need to do more than tweak a few numbers manually. 
To perform a scenario analysis using data validation drop-down boxes, follow 
these steps: 
 

1. Take the existing model and cut and paste the descriptions from column C to column 
F. You can do this by highlighting cells C6:C8, pressing Ctrl+X, selecting cell F6, and 
pressing Enter. 

2. The inputs in cells B3 to B8 are the active range that drives the model and will 
remain so. However, they need to become formulas that change depending on the 
drop-down box that you’ll create. 

3. Copy the range in column B across to columns C, D, and E. 
4. You can do this by highlighting B3:B8, pressing Ctrl+C, selecting cells C3:E3, and 

pressing Enter. These amounts will be the same for each scenario until we change 
them. 

5. In row 2 enter the titles Best Case, Base Case, and Worst Case, as shown in Figure 
8-1. 

Note that the formulas still link to the inputs in column B, as we can see by 
selecting cell C12 and pressing the F2 shortcut key as shown in figure 8-1. 

 

Figure 8-1: Setting up the model for scenario analysis 

6. Edit the inputs underneath each scenario. You can put whatever you think is likely, but 
in order to match the numbers to those in this example, enter the values as shown in 
Figure 8-2. Ignore column B for now. 

 

 

Figure 8-2: Inputs for scenario analysis 

7. Now you need to add the drop-down box at the top, which is going to drive 
your scenarios. It doesn’t really matter where exactly you put the drop-down 

box, but it should be in a location that’s easy to find, usually at the top of the 
page. 
8. In cell E1, enter the title Scenario:. 
9. Select cell F1, and change the formatting to input so that the user can see 
that this cell is editable. The easiest way to do this is to follow these steps: 

 Click one of the cells that are already formatted as an input, such as cell 
E3. 

 Press the Format Painter icon in the Clipboard section on the left-hand 
side of the Home tab., Your cursor will change to a paintbrush. 

 Select cell F1 to paste the formatting. 

<Tip> 
Format Painter is normally for single use. After you’ve selected the cell, the 
paintbrush will disappear from the cursor. If you want the Format Painter to 
become “sticky” and apply to multiple cells, double-click the icon when you 
select it from the Home tab. 
10. Now, in cell F1, select Data Validation from the Data Tools section of the 
Data tab. 
The Data Validation dialog box appears. 
11. On the Settings tab, change the Allow drop-down to List, use the mouse to 
select the range =$C$2:$E$2 (see Figure 8-3), and click OK. 

 

 

Figure 8-3: Creating the data validation drop-down scenarios 

 Click the drop-down box, which now appears next to cell F1, and select one 
of the scenarios (for example, Base Case). 

Applying formulas to scenarios 

The cells in column B are still driving the model, and these need to be replaced 
by formulas. Before you add the formulas, however, you should change the 
formatting of the cells in the range to show that they contain formulas, instead 
of hard-coded numbers. Follow these steps: 

1. Select cells B3:B8, and select the Fill Color from the Font group on 
the Home tab. 

2. Change the Fill Color to a white background. 

<Remember> 
It’s very important to distinguish between formulas and input cells in a model. 
You need to make it clear to any user opening the model that the cells in this 
range contain formulas and should not be overridden. 
 
Now you need to replace the hard-coded values in column B with formulas that 
will change as the drop-down box changes. You can do this using a number of 
different functions; an HLOOKUP, a nested IF statement, an IFS, and a SUMIF 
will all do the trick. Add the formulas by following these steps: 
 
 1. Select cell B3, and add a formula that will change the value 
depending on what is in cell F1. 
 
  Here is what the formula will be under the different options: 
 * =HLOOKUP($F$1,$C$2:$E$8,2,0) 
Note that with this solution, you need to change the row index number from 2 
to 3 and so on as you copy the formula down. Instead, you could use a ROW 
function in the third field like this: =HLOOKUP($F$1,$C$2:$E$8,ROW(A3)-1,0) 
 * =IF($F$1=$C$2,C3,IF($F$1=$D$2,D3,E3)) 
 * =IFS($F$1=$C$2,C3,$F$1=$D$2,D3,$F$1=$E$2,E3) 
 * =SUMIF($C$2:$E$2,$F$1,C3:E3) 

<Tip> 
  As always, there are several different options to choose from and 
the best solution is the one that is the simplest and easiest to understand. Any 
of these functions will produce exactly the same result, but in my opinion, 
having to change the row index number in the HLOOKUP is not robust, and 
adding the ROW may be confusing for a user. The nested IF statement is tricky 
to build and follow, and although the new IFS function is designed to make a 
nested IF function simpler, it’s still rather unwieldy. I find the SUMIF quite 
simple to build and follow, and it’s easy to expand if you need to add extra 
scenarios in the future. 

<Warning> 
Note that IFS is a new function that is only available with Office 365 and 
Excel 2016 or later installed. If you use this function and someone opens this 
model in a previous version of Excel, she can view the formula, but she won’t 
be able to edit it. 
 2. Copy the formula in cell B3 down the column. 

<Tip> 
By using an ordinary copy and paste, you’ll lose all your formatting. It’s 
important to retain the formatting of the model so that you can see at a 
glance which inputs are in dollar values, percentages, or customer numbers. 
Use Paste Formulas to retain the formatting. You can access it by copying the 
cell onto the clipboard, highlighting the destination range, right-clicking, and 
selecting the Paste Formulas icon to paste formulas only, and leave the 
formatting intact (see Figure 8-4). 

 

 

Figure 8-4: Using Paste Formulas to retain formatting. 

Now for the fun part! It’s time to test the scenario functionality in the model. 
Click cell F1, change the drop-down box, and watch the model outputs change 
as you toggle between the different scenarios, as shown in Figure 8-5. 

Figure 8-5: The completed scenario analysis 

ACTUAL VS BUDGET VARIANCE 

REPORT 

A typical Actual versus Budget report will have at 
least three columns; Actual, Budget, Variance – 
and possibly % Variance for each month.  This 
means at least 36 columns across the page, which is not a manageable or 
readable size for a good report.  A way around this is to create a concise, 
dynamic report into which actual and budgeted numbers can be compared 
and analysed quickly and efficiently.  This concise report will show only the 
current month depending on the month selected by the user. 

22. Open up the file called “Act vs Budget Variance Report”.   
23. On the “Summary” page, in cell A5, we need to create a linked drop-down 

which will enable the user to select the current month.  Do this by: 
24. Go to the “Actual” page and create a named range 

called “months” which includes all the dates in row 4. 
25. Go back to the “Summary” page, and create a 

drop-down box. Link it to dates using the named 
range. 

26. Name cell A5 as “curr_month”, and select February 
just as an example to get your model working. 

27. On the “P&L” page, we need to use a SUMIF to 
populate the Actual column in column C.   In cell C6, 
create a SUMIF referencing the “Actuals” sheet 
where the periods are the range, the criteria is the 
current month, and the sum range is the 
corresponding row in the “Budget” sheet. 

If you have created the named range for your months then your wizard in cell 
C6 should look like this, and the result should be $222,604: 

 

 

28. The formula should be: =SUMIF(months,C$4,Actuals!$C6:$N6) Don’t forget to 
use absolute referencing! 

29. Copy the formula all the way down to the bottom. 
30. Now, do exactly the same thing to populate the “Budget” column, by pulling the 

data from the Budget page.  Your formula in cell D6 should be 
=SUMIF(Budget!$C$4:$N$4,D$4,Budget!C6:N6) 

See page 107 of “Using 
Excel for Business and 
Financial Modelling” for 
how to create a Named 
Range and  page 205 for 
how to create  data 
validation drop-down list. 

See page 104 of “Using 
Excel for Business and 
Financial Modelling” for 
more information on 
using cell referencing. 

Note that a VLOOKUP could also be used to complete this task, but a SUMIF is 
simpler in this case.  We’ll come back to the variance in column E later.  Let’s 
move on to the Year to Date columns in F & G. 

Creating a Year to Date (YTD) Formula 

We need to identify on the “Budget” and “Actual” pages which months have 
already passed, and therefore should be included in the Year to Date (YTD) 
formula.  The easiest way to do this is to create an IF statement which will 
return the word “actual” if we already have actuals for this month or “budget” 
if we do not. 

1. Go to the “Budget” page and in the blank row 5 under the periods in cell C5, enter your 
IF statement.  

2. Your formula should look like this: =IF(C4<=curr_month,1,0). Because we’ve got 
February selected, January and February should be showing a 1 and the other months a 
zero.  This will automatically change as we move through the months. 

3. Copy this formula all the way across the row, and repeat for the “Actual” page.   
4. Now go back to the “P&L” page and in the Actual column in cell F6, create another 

SUMIF which will pick up ONLY the budget for the months which have already passed. 
5. Your wizard should look like this: 

 

6. The formula should be: =SUMIF(Actuals!$C$5:$N$5,1,Actuals!C6:N6), and the result 
$478,453. Don’t forget your absolute referencing.   

7. Copy this formula all the way down to the bottom of the report.  Instead of copying the 
entire column and having to recreate the SUM cells, use the shortcuts Control-c and then 
Control-v repeatedly to copy down the cells instead. 

8. Now go to column G and repeat the procedure for the budget as well. 
9. Your formula in cell G6 should be =SUMIF(Budget!$C$5:$N$5,1,Budget!C6:N6).  Copy 

it down. 

Adding Variances 
In typical variance formulas, better than budget must show as a positive 
value, and worse than budget must show as a negative value.  Therefore 
there is a different formula for revenue and expenses.   

1. For Revenue and profit, our formula will be ACTUAL – BUDGET, and for expenses it will 
be BUDGET – ACTUAL. 

2. In columns E and G insert the variance formulas. 
3. In cell E6 (the revenue row) your formula should be:  =C6-D6.   

Repeat for column G. 
4. In cell E10 (the revenue row) your formula should be:  =D10-C10.  Repeat for column G. 
5. Alternatively, you could create a more intelligent formula which looks at the first 

character of the account code to determine whether it is a revenue or expense item.  If 
you choose to do this, then your formula in cell E6 would be =IF(LEFT(A6,1)="4",C6-

D6,D6-C6), and this follows better financial modelling practice because it means the 
formulas can be consistent all the way down the page, rather than changing for 
expenses. 

Actual / Budget 
It would be useful to know how we are tracking for the full year, therefore we 
will insert a full year Budget and Actual / Budget columns. 

1. In cell I6, sum the entire budget with the formula =SUM(Budget!C6:N6) and copy it all 
the way down. 

2. In cell J6, add another SUMIF formula which adds only the budget columns which are in 
the future, and add the actual amounts to it.  The columns which are in the future have a 
zero in row 5.  Your formula should be: 
=SUMIF(Budget!$C$5:$K$5,0,Budget!C6:K6)+F6. 

3. Copy this all the way down to the bottom of the report.Your Actual versus Budget P&L 
report is now complete.  Go back to the “Summary” page, and practice changing the 
month from the drop-downs and see the numbers change. 

 

Adding a Trendline with Sparklines 
It’s useful to see on the Summary page how 
the full year trends.  There isn’t enough data 
to show the actuals, but the budget has a 
whole year’s worth of data we can show on 
the summary. 

1. Go to the Budget page and for each summary line, starting from cell O7, create a 
sparkline. 

2. Copy this down to each summary line as shown below 
3. Change the weight and colour of the sparkline if desired 
4. Cut and paste each sparkline to the corresponding cell on the Summary page. 

 

See page 195 of “Using 
Excel for Business and 
Financial Modelling” for 
instructions on how to 
create a sparkline. 

